

ΔΙΚΤΥΑ ΕΠΙΚΟΙΝΩΝΙΩΝ

Εργαστηριακή Άσκηση 8

Ασύρματα Τοπικά Δίκτυα IEEE 802.11

1. Εισαγωγή

Σε αυτή την άσκηση θα μελετηθεί η συμπεριφορά ασυρμάτων τοπικών δικτύων που ακολουθούν το πρότυπο IEEE 802.11. Η αρχιτεκτονική, η στοίβα και η λειτουργία πρωτοκόλλων, η δομή των πλαισίων και οι υπηρεσίες του προτύπου αυτού περιγράφονται αναλυτικά στο βιβλίο «Δίκτυα Υπολογιστών» (A. Tanenbaum), στο Κεφάλαιο 4.

Αρχικά θα γραφεί σε γλώσσα tcl ένα script για τον NS2, το οποίο θα προσομοιώνει ένα ασύρματο τοπικό δίκτυο IEEE 802.11 με δύο κόμβους, όπου ο ένας στέλνει πακέτα στον άλλον. Από την προσομοίωση θα προκύψει αρχείο ίχνους (trace file), από την ανάλυση του οποίου θα προκύψουν κάποια μέτρα επίδοσης του δικτύου. Συγκεκριμένα θα μετρηθεί η μέση τιμή της καθυστέρησης ανά πακέτο, καθώς και η χρησιμοποίηση του διαύλου. Ακολούθως, θα γίνει προσομοίωση με περισσότερους κόμβους και θα ζητηθεί η μέτρηση αντιστοίχων μεγεθών.

Στην ενότητα 2 παρουσιάζεται το πρότυπο IEEE 802.11 εστιάζοντας στα σημεία που υπεισέρχονται στην παρούσα εργαστηριακή άσκηση. Στην ενότητα 3 παρουσιάζονται οι λεπτομέρειες του σεναρίου προσομοίωσης και στην ενότητα 4 εκτελείται σειρά προσομοιώσεων, όπου μελετάται η συμπεριφορά του IEEE 802.11.

2. Πρότυπο IEEE 802.11

Το πρότυπο IEEE 802.11 υποστηρίζει λειτουργία με δύο τρόπους:

1. Με παρουσία σταθμού βάσης.
2. Με απουσία σταθμού βάσης.


Στην πρώτη περίπτωση όλες οι επικοινωνίες πρέπει να περνούν από το σταθμό βάσης, ο οποίος ονομάζεται **σημείο πρόσβασης** (access point) στην ορολογία του 802.11. Στη δεύτερη περίπτωση οι υπολογιστές απλώς μεταδίδουν απευθείας ο ένας στον άλλον. Αυτός ο τρόπος λειτουργίας ονομάζεται συχνά **δίκτυωση ad hoc** (ad hoc networking).

Η εργαστηριακή αυτή άσκηση μελετά τη δεύτερη περίπτωση λειτουργίας του προτύπου IEEE 802.11 και γίνεται η υπόθεση ότι όλοι οι ραδιοπομποί έχουν την ίδια σταθερή εμβέλεια. Όταν ένας δέκτης βρίσκεται μέσα στην εμβέλεια δύο ενεργών πομπών, το σήμα που προκύπτει είναι γενικά παραμορφωμένο και άχρηστο – με άλλα λόγια, στην ανάλυση που ακολουθεί δεν θα ασχοληθούμε με συστήματα τύπου CDMA. Το γεγονός ότι εν γένει στα ασύρματα LAN, ο κάθε σταθμός δεν βρίσκεται απαραίτητως μέσα στην εμβέλεια όλων των άλλων σταθμών οδηγεί σε ποικιλία προβλημάτων. Επιπλέον, για τα ασύρματα LAN εντός κτιρίων, η παρουσία των τοίχων, ή αντίστοιχων εμποδίων ανάμεσα στους σταθμούς μπορεί να έχει μεγάλο αντίκτυπο στην πραγματική εμβέλεια του κάθε σταθμού.

Μια απλοϊκή προσέγγιση στη χρήση ενός ασύρματου LAN θα μπορούσε να είναι η προσπάθεια χρήσης του CSMA: να ακούμε για άλλες μεταδόσεις και να μεταδίδουμε μόνο αν κανείς άλλος δεν το κάνει. Το πρωτόκολλο αυτό όμως δεν είναι πραγματικά κατάλληλο για την περίπτωση, επειδή αυτό που έχει σημασία είναι οι παρεμβολές στον παραλήπτη, και όχι στον αποστολέα.

Ένα πρώιμο πρωτόκολλο που σχεδιάστηκε για ασύρματα LAN είναι το **πρωτόκολλο Πολλαπλής Πρόσβασης με Αποφυγή Συγκρούσεων** ή **MACA** (Multiple Access with Collision Avoidance, Karn 1990). Η βασική ιδέα πίσω από το πρωτόκολλο αυτό είναι να «διεγείρει» ο αποστολέας τον παραλήπτη

έτσι ώστε ο παραλήπτης να στείλει ένα μικρό πλαίσιο με αποτέλεσμα οι κοντινοί σε αυτόν σταθμοί να εντοπίσουν τη μετάδοση και να αποφύγουν τις μεταδόσεις κατά τη διάρκεια του επερχόμενου (μεγάλου) πλαισίου δεδομένων. Το πρωτόκολλο MACA απεικονίζεται στο Σχήμα 1.



Σχήμα 1 - Το πρωτόκολλο MACA. (a) Ο Α στέλνει ένα πλαίσιο RTS στον Β. (b) Ο Β ανταποκρίνεται στον Α με ένα πλαίσιο CTS.

Ας δούμε τώρα πώς ο Α στέλνει ένα πλαίσιο στον Β. Ο Α ξεκινά στέλνοντας ένα πλαίσιο **Αίτησης Αποστολής** ή **RTS** (Request To Send) στον Β, όπως φαίνεται στο Σχήμα 1 (a). Αυτό το μικρό πλαίσιο (30 byte) περιέχει το μέγεθος του πλαισίου δεδομένων που θα ακολουθήσει αργότερα. Ο Β ανταποκρίνεται στη συνέχεια με ένα πλαίσιο **Έγκρισης Αποστολής** ή **CTS** (Clear to Send), όπως φαίνεται στο Σχήμα 1 (b). Το πλαίσιο CTS περιέχει το μήκος των δεδομένων (αντιγράφεται από το πλαίσιο RTS). Με λήψη του πλαισίου CTS, ο Α αρχίζει τη μετάδοση.

Ας δούμε τώρα πώς αντιδρούν οι σταθμοί που ακούν κάποιο από αυτά τα πλαίσια. Ένας σταθμός που ακούει το RTS βρίσκεται προφανώς κοντά στον Α, και θα πρέπει να παραμείνει σιωπηλός για αρκετό χρόνο έτσι ώστε να επιστραφεί το πλαίσιο CTS στον Α χωρίς σύγκρουση. Κάθε σταθμός που ακούει το CTS βρίσκεται προφανώς κοντά στον Β και θα πρέπει να παραμείνει σιωπηλός κατά την επερχόμενη μετάδοση δεδομένων, το μήκος της οποίας μπορεί να προσδιοριστεί μέσω του πλαισίου CTS.

Στο Σχήμα 1, ο C βρίσκεται εντός της εμβέλειας του Α αλλά εκτός της εμβέλειας του Β. Κατά συνέπεια, ακούει το πλαίσιο RTS από τον Α αλλά όχι το πλαίσιο CTS από τον Β. Με την προϋπόθεση ότι δεν θα παρεμβληθεί με το CTS, είναι ελεύθερος να μεταδώσει ενώ στέλνεται το πλαίσιο δεδομένων. Αντιθέτως, ο D βρίσκεται εντός της εμβέλειας του Β αλλά όχι και του Α. Δεν ακούει το RTS αλλά ακούει το CTS. Η ακρόαση του CTS τον ενημερώνει ότι βρίσκεται κοντά σε ένα σταθμό που πρόκειται να λάβει ένα πλαίσιο, έτσι αποφεύγει να στείλει οτιδήποτε μέχρι τον αναμενόμενο χρόνο ολοκλήρωσης του πλαισίου αυτού. Ο σταθμός Ε ακούει και τα δύο μηνύματα ελέγχου και, όπως και D, θα πρέπει να παραμείνει σιωπηλός μέχρι να ολοκληρωθεί τα πλαίσια δεδομένων.

Παρά τις προφυλάξεις αυτές, μπορούν και πάλι να συμβούν συγκρούσεις. Για παράδειγμα, ο Β και ο C μπορούν να στείλουν ταυτόχρονα και οι δύο πλαίσια RTS στον Α. Τα πλαίσια αυτά θα συγκρουστούν και θα χαθούν. Σε περίπτωση σύγκρουσης, ο πομπός όπου συνέβη η αποτυχία (δηλαδή, ένας πομπός που δεν ακούει το πλαίσιο CTS μέσα στο αναμενόμενο χρονικό διάστημα) θα περιμένει ένα τυχαίο χρονικό διάστημα και θα ξαναδοκιμάσει αργότερα. Ο αλγόριθμος που χρησιμοποιείται είναι η δυαδική εκθετική οπισθοχώρηση, η οποία χρησιμοποιείται και στο Ethernet.


Όπως αναφέρθηκε στην αρχή, το 802.11 υποστηρίζει δύο καταστάσεις λειτουργίας. Η πρώτη δεν χρησιμοποιεί κάποιο είδος κεντρικού ελέγχου (οπότε από αυτή την άποψη, είναι παρόμοια με το Ethernet) και ονομάζεται **Κατανεμημένη Λειτουργία Συντονισμού** ή **DCF** (Distributed Coordination Function). Η άλλη χρησιμοποιεί το σταθμό βάσης για τον έλεγχο όλων των δραστηριοτήτων στην αντίστοιχη κυψέλη του και ονομάζεται **Σημειακή Λειτουργία Συντονισμού** ή **PCF** (Point Coordination Function). Όλες οι υλοποιήσεις πρέπει να υποστηρίζουν την DCF, ενώ η PCF είναι

προαιρετική. Θα εξετάσουμε μόνο την πρώτη από τις δύο καταστάσεις λειτουργίας παρακάτω, καθώς αυτή αποτελεί το αντικείμενο της παρούσας εργαστηριακής άσκησης.

Όταν χρησιμοποιείται η DCF, το 802.11 χρησιμοποιεί ένα πρωτόκολλο που ονομάζεται CSMA με Αποφυγή Συγκρούσεων ή CSMA/CA (CSMA with Collision Avoidance). Στο πρωτόκολλο αυτό χρησιμοποιείται ανίχνευση τόσο του φυσικού όσο και του εικονικού καναλιού. Το CSMA/CA υποστηρίζει δύο μεθόδους λειτουργίας. Στην πρώτη μέθοδο, όταν ένας σταθμός θέλει να μεταδώσει ανιχνεύει το κανάλι. Αν είναι αδρανές αρχίζει να μεταδίδει. Καθώς μεταδίδει δεν ανιχνεύει το κανάλι, αλλά στέλνει ολόκληρο το πλαίσιο του, το οποίο μπορεί βέβαια να καταστραφεί στον παραλήπτη λόγω παρεμβολών εκεί. Αν το κανάλι είναι απασχολημένο, ο αποστολέας αναβάλλει τη μετάδοση μέχρι το κανάλι να γίνει αδρανές, και τότε αρχίζει να μεταδίδει. Αν συμβεί μια σύγκρουση, οι σταθμοί που συγκρούστηκαν αναμένουν ένα τυχαίο χρονικό διάστημα χρησιμοποιώντας τον αλγόριθμο δυαδικής εκθετικής οπισθοχώρησης του Ethernet και ξαναδοκιμάζουν αργότερα.

Ο άλλος τρόπος λειτουργίας του CSMA/CA βασίζεται στον MACAW και χρησιμοποιεί ανίχνευση του εικονικού καναλιού, όπως φαίνεται στο Σχήμα 2. Στο παράδειγμα αυτό, ο A θέλει να στείλει στον B. Ο C εντός της εμβέλειας του A (και πιθανόν εντός της εμβέλειας του B, αυτό όμως δεν έχει σημασία). Ο D είναι ένας σταθμός εντός της εμβέλειας του B αλλά όχι εντός της εμβέλειας του A.

Το πρωτόκολλο αρχίζει όταν ο A αποφασίσει ότι θέλει να στείλει δεδομένα στον B. Ξεκινά στέλνοντας ένα πλαίσιο RTS στον B, ζητώντας άδεια να του στείλει ένα πλαίσιο. Όταν ο B λάβει την αίτηση αυτή, μπορεί να αποφασίσει να παραχωρήσει τη ζητούμενη άδεια, οπότε επιστρέφει ένα πλαίσιο CTS. Με τη λήψη του CTS, ο A στέλνει το πλαίσιο του και ξεκινά ένα χρονόμετρο επιβεβαίωσης. Αφού λάβει ορθά το πλαίσιο δεδομένων, ο B απαντά με ένα πλαίσιο επιβεβαίωσης, ολοκληρώνοντας την ανταλλαγή. Αν το χρονόμετρο επιβεβαίωσης του A λήξει πριν φτάσει σε αυτόν η επιβεβαίωση, ολόκληρο το πρωτόκολλο εκτελείται ξανά.



Σχήμα 2 - Ανίχνευση εικονικού καναλιού με το CSMA/CA.


Ας εξετάσουμε τώρα την ανταλλαγή αυτή από την οπτική γωνία των C και D. Ο C βρίσκεται εντός της εμβέλειας του A, έτσι μπορεί να λάβει το πλαίσιο RTS. Αν το λάβει αντιλαμβάνεται ότι σε λίγο κάποιος θα στείλει δεδομένα, έτσι για το καλό όλων αποφεύγει να μεταδώσει οτιδήποτε μέχρι να ολοκληρωθεί η ανταλλαγή. Από τις πληροφορίες που παρέχονται στην αίτηση RTS μπορεί να εκτιμήσει πόσο χρόνο θα πάρει η ανταλλαγή, συμπεριλαμβανομένης της τελικής επιβεβαίωσης, έτσι ενεργοποιεί για τον εαυτό του ένα σήμα ότι το εικονικό κανάλι είναι απασχολημένο, γεγονός που σημειώνεται στο Σχήμα 2 ως Διάνυσμα Εκχώρησης Δικτύου ή NAV (Network Allocation Vector). Ο D δεν ακούει το μήνυμα RTS, ακούει όμως το CTS, έτσι ενεργοποιεί και αυτός από την πλευρά του το σήμα NAV. Σημειώστε ότι τα σήματα NAV δεν μεταδίδονται. Είναι απλώς εσωτερικές υπενθυμίσεις ότι ο σταθμός πρέπει να παραμείνει σιωπηλός για μια συγκεκριμένη χρονική περίοδο.

Όλα τα παραπάνω ισχύουν για την κατάσταση λειτουργίας DCF του 802.11. Σε αυτή την κατάσταση λειτουργίας δεν υπάρχει κεντρικός έλεγχος και οι σταθμοί ανταγωνίζονται για το κανάλι, ακριβώς όπως και στο Ethernet.

3. Σενάριο προσομοίωσης

3.1 Παράμετροι Προσομοίωσης, Τοπολογία Δικτύου και Ορισμός Αρχείων Ιχνών

Στο πλαίσιο της προσομοίωσης θα υλοποιηθεί μια ασύρματη δικτυακή τοπολογία όπως φαίνεται για παράδειγμα στο Σχήμα 3.



Σχήμα 3 – Τοπολογία ασυρμάτου τοπικού δικτύου.

Στην τοπολογία αυτή, ορίζουμε επίπεδο πλέγμα διαστάσεων $opt(gridx) \times opt(gridy)$, όπου όλοι οι κόμβοι είναι τοποθετημένοι στο μέσο του άξονα y και σε απόσταση $opt(distx)$ μεταξύ τους στον άξονα x , συμμετρικά ως προς το κέντρο του. Στην οριζόντια διάταξη που προκύπτει, ο πρώτος κόμβος από αριστερά παράγει $opt(cbrmaxph)$ πακέτα UDP κίνησης CBR με προορισμό τον τελευταίο κόμβο από τα δεξιά. Στο τμήμα του κώδικα που ακολουθεί, καθορίζονται οι παράμετροι της προσομοίωσης. Οι σημαντικότερες από αυτές τις παραμέτρους είναι το πλήθος των ασύρματων κόμβων $opt(nn)$, η οριζόντια απόσταση μεταξύ δύο διαδοχικών κόμβων $opt(distx)$, το μέγεθος των πακέτων UDP $opt(udpsize)$, η χρονική στιγμή έναρξης της κίνησης CBR $opt(cbrstart)$, η διάρκεια της προσομοίωσης $opt(simstop)$. Αυτές οι παράμετροι πρέπει να αλλάζουν σε κάθε προσομοίωση, όπως περιγράφεται στη συνέχεια.

```
# =====
# Ορισμός παραμέτρων προσομοίωσης
# =====

set opt(chan)          Channel/WirelessChannel ;# Τύπος καναλιού
set opt(prop)          Propagation/TwoRayGround ;# Μοντέλο ραδιομετάδοσης
set opt(ant)           Antenna/OmniAntenna      ;# Τύπος κεραίας
set opt(ll)            LL                          ;# Τύπος επιπέδου σύνδεσης
set opt(ifq)           Queue/DropTail/PriQueue   ;# Τύπος ουράς
set opt(ifqlen)        20                        ;# Μέγιστος αριθμός πακέτων
                                              ;# στην ουρά

set opt(netif)         Phy/WirelessPhy          ;# Τύπος δικτυακής επαφής
set opt(mac)           Mac/802_11               ;# Πρωτόκολλο MAC
set opt(rp)            AODV                     ;# Πρωτόκολλο δρομολόγησης
set opt(nn)            2                       ;# Αριθμός κόμβων
set opt(gridx)         200                     ;# Μήκος πλέγματος (m)
```

```

set opt(gridy)          200          ;# Πλάτος πλέγματος (m)
set opt(distx)          10           ;# Οριζόντια απόσταση μεταξύ
                                     ;# διαδοχικών κόμβων (e)m

set opt(udpssize)       1000         ;# Μέγεθος UDP (byte)
set opt(cbrpssize)      1000         ;# Μέγεθος πακέτου CBR (byte)
set opt(cbrpinterval)  0.02         ;# Χρόνος μεταξύ διαδοχικών
                                     ;# πακέτων CBR (sec)

set opt(cbrmaxpn)       300          ;# Μέγιστος αριθμός πακέτων CBR
set opt(cbrstart)       0.25         ;# Χρόνος εκκίνησης CBR
set opt(simstop)        8.0          ;# Χρόνος λήξης προσομοίωσης

$opt(mac) set basicRate_ 1Mb
$opt(mac) set dataRate_ 11Mb

# Δημιουργία αντικειμένου προσομοίωσης
set ns [new Simulator]

# Ορισμός αρχείων ιχνών
set tf [open lab8.tr w]
$ns trace-all $tf

set nf [open lab8.nam w]
$ns namtrace-all-wireless $nf $opt(gridx) $opt(gridy)

# Συνάρτηση τερματισμού
proc finish {} {
    global ns tf nf
    $ns flush-trace
    close $tf
    close $nf
    exit 0
}

# Δημιουργία αντικειμένου τοπογραφίας
set topo [new Topography]
# Επίπεδο πλέγμα ( $opt(gridx) x $opt(gridy) ) m^2
$topo load_flatgrid $opt(gridx) $opt(gridy)

# Δημιουργία αντικειμένου GOD (General Operations Director) που αποθηκεύει τον
# συνολικό αριθμό κόμβων και έναν πίνακα με τον ελάχιστο αριθμό βημάτων (hops)
# μεταξύ των κόμβων
create-god $opt(nn)

```

```

# Παραμετροποίηση κόμβων
$ns node-config -adhocRouting $opt(rp) \
  -llType $opt(ll) \
  -macType $opt(mac) \
  -ifqType $opt(ifq) \
  -ifqLen $opt(ifqlen) \
  -antType $opt(ant) \
  -propType $opt(prop) \
  -phyType $opt(netif) \
  -channel [new $opt(chan)] \
  -topoInstance $topo \
  -agentTrace ON \
  -routerTrace ON \
  -macTrace OFF \
  -movementTrace OFF

#
# Οι κόμβοι τοποθετούνται στον άξονα x συμμετρικά ως προς το μέσο του,
# ενώ η απόσταση μεταξύ δύο διαδοχικών κόμβων είναι ίση με $opt(distx)
#

# Ορισμός θέσης πρώτου κόμβου στον άξονα x
set posx(0) [expr $opt(gridx)/2.0-(($opt(nn)-1)/2.0)*$opt(distx)]

# Δημιουργία κόμβων και ορισμός συντεταγμένων τους στο επίπεδο πλέγμα
for {set i 0} {$i < $opt(nn)} {incr i} {
  # Δημιουργία κόμβου
  set n($i) [$ns node]
  # Απενεργοποίηση τυχαίας κίνησης κόμβου
  $n($i) random-motion 0
  # Ορισμός θέσης κόμβου στον άξονα x
  $n($i) set X_ [expr $posx(0) + $i*$opt(distx)]
  # Ορισμός θέσης κόμβου στον άξονα y
  $n($i) set Y_ [expr $opt(gridy)/2.0]
  # Ορισμός θέσης κόμβου στον άξονα z
  $n($i) set Z_ 0.0
  set node_x [$n($i) set X_]
  set node_y [$n($i) set Y_]
  puts "node($i) at position ($node_x, $node_y)"
}

```

2.2 Γεννήτρια Κίνησης

Για τη μετάδοση δεδομένων μεταξύ των ακραίων κόμβων χρησιμοποιούμε το πρωτόκολλο UDP. Η κίνηση αυτή ξεκινά τη χρονική στιγμή *opt(cbrstart)*.

```
# Ορισμός μεγέθους UDP datagram
Agent/UDP set packetSize_ $opt(udpssize)
set agent [new Agent/UDP]

# Ορισμός πρώτου κόμβου ως αποστολέα κίνησης CBR
set cbr [new Application/Traffic/CBR]
$cbr set packetSize_ $opt(cbrpsize)
$cbr set interval_ $opt(cbrpinterval)
$cbr set maxpkts_ $opt(cbrmaxpn)
$cbr attach-agent $agent
$ns attach-agent $n(0) $agent

# Ορισμός τελευταίου κόμβου ως παραλήπτη
set sink [new Agent/LossMonitor]
$ns attach-agent $n([expr $opt(nn)-1]) $sink

# Σύνδεση αποστολέα - παραλήπτη
$ns connect $agent $sink
```

2.3 Εκτέλεση του σεναρίου

Τέλος, ορίζουμε τα γεγονότα της προσομοίωσης:

```
# Αρχικοποίηση κινητών κόμβων
for {set i 0} {$i < $opt(nn)} {incr i} {
    $ns initial_node_pos $n($i) 10
}

# Ορισμός ετικέτας πρώτου κόμβου
$ns at 0.0 "$n(0) label Sender"
# Ορισμός ετικέτας τελευταίου κόμβου
$ns at 0.0 "$n([expr $opt(nn)-1]) label Receiver"

# Χρόνος έναρξης κίνησης CBR $opt(cbrstart) δευτερόλεπτων.
$ns at $opt(cbrstart) "$cbr start"

# Reset όλων των κόμβων
for {set i 0} {$i < $opt(nn)} {incr i} {
    $ns at $opt(simstop) "$n($i) reset";
}
```

```
# Τερματισμός
$ns at $opt(simstop) "finish"

$ns run
```

Με βάση τα παραπάνω δημιουργούμε το αρχείο **lab8.tcl**, ενώ για την εκτέλεση της προσομοίωσης χρησιμοποιούμε την εντολή **ns lab8.tcl**. Μετά την εκτέλεση αυτής της εντολής θα πρέπει να έχουν δημιουργηθεί τα αρχεία **lab8.tr** και **lab8.nam**. Με την εντολή **nam lab8.nam** μπορούμε να δούμε την τοπολογία του δικτύου καθώς και την κίνηση που έχει δημιουργηθεί.

Το αρχείο **lab8.tr** περιέχει πληροφορίες για όλα τα γεγονότα που συνέβησαν κατά την προσομοίωση. Στην επόμενη ενότητα θα εξηγηθεί πώς αναλύονται αυτά τα δεδομένα.

3. Ανάλυση αρχείου ίχνους (trace file)

Αφού έχουμε δημιουργήσει το σενάριο προσομοίωσης, το έχουμε εκτελέσει και έχουμε δημιουργήσει τα αρχεία αποτελεσμάτων, τα αρχεία αυτά πρέπει να αναλυθούν ώστε να πάρουμε τις πληροφορίες που θέλουμε.

3.1 Μορφή αρχείου ίχνους (trace file)

Το αρχείο **lab8.tr** που δημιουργήθηκε προηγουμένως περιέχει πληροφορίες της μορφής:

```
s 0.250000000 _0_ AGT --- 0 cbr 1000 [0 0 0 0] ----- [0:0 31:0 32 0] [0] 0 0
r 0.250000000 _0_ RTR --- 0 cbr 1000 [0 0 0 0] ----- [0:0 31:0 32 0] [0] 0 0
s 0.250000000 _0_ RTR --- 0 AODV 48 [0 0 0 0] ----- [0:255 -1:255 30 0] [0x2 1 1 [31 0] [0 4]] (REQUEST)
r 0.250900067 _1_ RTR --- 0 AODV 48 [0 ffffffff 0 800] ----- [0:255 -1:255 30 0] [0x2 1 1 [31 0] [0 4]] (REQUEST)
r 0.250900133 _2_ RTR --- 0 AODV 48 [0 ffffffff 0 800] ----- [0:255 -1:255 30 0] [0x2 1 1 [31 0] [0 4]] (REQUEST)
r 0.250900200 _3_ RTR --- 0 AODV 48 [0 ffffffff 0 800] ----- [0:255 -1:255 30 0] [0x2 1 1 [31 0] [0 4]] (REQUEST)
r 0.250900267 _4_ RTR --- 0 AODV 48 [0 ffffffff 0 800] ----- [0:255 -1:255 30 0] [0x2 1 1 [31 0] [0 4]] (REQUEST)
r 0.250900333 _5_ RTR --- 0 AODV 48 [0 ffffffff 0 800] ----- [0:255 -1:255 30 0] [0x2 1 1 [31 0] [0 4]] (REQUEST)
r 0.250900400 _6_ RTR --- 0 AODV 48 [0 ffffffff 0 800] ----- [0:255 -1:255 30 0] [0x2 1 1 [31 0] [0 4]] (REQUEST)
r 0.250900467 _7_ RTR --- 0 AODV 48 [0 ffffffff 0 800] ----- [0:255 -1:255 30 0] [0x2 1 1 [31 0] [0 4]] (REQUEST)
r 0.250900533 _8_ RTR --- 0 AODV 48 [0 ffffffff 0 800] ----- [0:255 -1:255 30 0] [0x2 1 1 [31 0] [0 4]] (REQUEST)
r 0.250900600 _9_ RTR --- 0 AODV 48 [0 ffffffff 0 800] ----- [0:255 -1:255 30 0] [0x2 1 1 [31 0] [0 4]] (REQUEST)
r 0.250900667 _10_ RTR --- 0 AODV 48 [0 ffffffff 0 800] ----- [0:255 -1:255 30 0] [0x2 1 1 [31 0] [0 4]] (REQUEST)
r 0.250900733 _11_ RTR --- 0 AODV 48 [0 ffffffff 0 800] ----- [0:255 -1:255 30 0] [0x2 1 1 [31 0] [0 4]] (REQUEST)
r 0.250900800 _12_ RTR --- 0 AODV 48 [0 ffffffff 0 800] ----- [0:255 -1:255 30 0] [0x2 1 1 [31 0] [0 4]] (REQUEST)
s 0.251042180 _2_ RTR --- 0 AODV 48 [0 ffffffff 0 800] ----- [2:255 -1:255 29 0] [0x2 2 1 [31 0] [0 4]] (REQUEST)
```

Κάθε γραμμή του αρχείου αυτού αντιστοιχεί σε ένα γεγονός που συνέβη κατά τη διάρκεια της προσομοίωσης. Ο πρώτος χαρακτήρας κάθε γραμμής υποδηλώνει το είδος του γεγονότος που συνέβη. Ο χαρακτήρας “s” σημαίνει επιτυχημένη αποστολή πακέτου, ενώ ο χαρακτήρας “r” σημαίνει επιτυχημένη λήψη πακέτου, και ο χαρακτήρας “D” σημαίνει απόρριψη πακέτου.

Η δεύτερη λέξη της κάθε γραμμής είναι η χρονική στιγμή κατά την οποία συνέβη το γεγονός που καταγράφεται. Η τρίτη λέξη περιγράφει τον κόμβο στον οποίον αναφέρεται το γεγονός. Η τέταρτη λέξη δηλώνει το επίπεδο όπου λαμβάνει χώρα το γεγονός της γραμμής (π.χ. AGT → agent, RTR → router). Η έκτη λέξη αναπαριστά το μοναδικό αριθμό ακολουθίας που αποδίδει το NS2 στα πακέτα. Η έβδομη λέξη δηλώνει το είδος του πακέτου και η όγδοη λέξη περιγράφει το μέγεθος του πακέτου (συμπεριλαμβάνονται οι επικεφαλίδες UDP και IP). Τα υπόλοιπα πεδία δεν χρησιμοποιούνται στην ανάλυση των αποτελεσμάτων της προσομοίωσης που ακολουθεί.

3.2 Ανάλυση με τη γλώσσα awk

Για να μπορούμε να υπολογίσουμε την χρησιμοποίηση (utilisation) του καναλιού, όταν χρησιμοποιείται το πρωτόκολλο IEEE 802.11, πρέπει να μετρήσουμε την ποσότητα των δεδομένων που ελήφθησαν κατά τη διάρκεια της προσομοίωσης. Στη συνέχεια ο αριθμός αυτός θα διαιρεθεί με τη διάρκεια της μετάδοσης του συνόλου των πακέτων, ώστε να υπολογιστεί ο ρυθμός διέλευσης δεδομένων πάνω από το ασύρματο κανάλι (throughput), και αυτή η τιμή θα διαιρεθεί με τον ονομαστικό ρυθμό μετάδοσης του καναλιού (11 Mbps) ώστε να προκύψει η χρησιμοποίηση (utilisation).

Για κάθε πακέτο που λαμβάνει ο αποδέκτης, η ποσότητα των δεδομένων αυξάνεται κατά το μέγεθος του πακέτου. Συνεπώς πρέπει να εντοπίσουμε τις γραμμές του αρχείου ίχνους που φανερώνουν ορθή λήψη πακέτου UDP της κίνησης CBR και επιπέδου agent. Αυτές οι γραμμές αρχίζουν με τον χαρακτήρα "r" και περιλαμβάνουν τις λέξεις *cbr* και *AGT*. Για κάθε τέτοια γραμμή που εντοπίζεται, η μεταβλητή *packets* (αριθμός πακέτων που ελήφθησαν) αυξάνεται κατά ένα, ενώ η μεταβλητή *data* (πλήθος δεδομένων που ελήφθησαν) αυξάνεται κατά το μέγεθος του πακέτου (το οποίο βρίσκεται στην όγδοη λέξη κάθε γραμμής του αρχείου ίχνους). Έτσι καταγράφεται η ποσότητα των δεδομένων που ελήφθησαν.

Ορίζουμε και διάφορες επιπλέον μεταβλητές, όπως οι ακόλουθες: *t_start* (χρόνος που έφυγε πρώτο πακέτο κίνησης CBR από αποστολέα), *t_end* (χρόνος που έφτασε τελευταίο πακέτο κίνησης CBR σε παραλήπτη), *rt_start* (χρόνος που έφυγε πρώτο AODV πακέτο από αποστολέα), *rt_end* (χρόνος που έφτασε τελευταίο AODV πακέτο σε αποστολέα).

Η μέση καθυστέρηση μετάδοσης πακέτου προκύπτει αν αθροιστούν οι χρόνοι μετάδοσης κάθε πακέτου κίνησης CBR και ακολούθως διαιρεθούν με το συνολικό πλήθος των πακέτων αυτών που μεταδόθηκαν επιτυχώς. Ο χρόνος μετάδοσης κάθε τέτοιου πακέτου προκύπτει αν από το χρόνο λήψης αυτού του πακέτου κίνησης CBR από τον παραλήπτη αφαιρεθεί ο χρόνος αποστολής του από τον αποστολέα.

Ο κώδικας awk για την εκτέλεση της απαραίτητης ανάλυσης είναι ο ακόλουθος:

```
BEGIN {
    data = 0;
    packets = 0;
    sumDelay = 0;

    # Do we need to fix the decimal mark?
    if (sprintf(sqrt(2)) ~ /\,/) dmfix = 1;
}

{
    # Apply dm fix if needed
    if (dmfix) sub(/\./, ",", $0);
}

/^s/&&/AGT/&&/cbr/ {
    if (t_start == "")
        t_start = $2;
```

```

        sendtimes[$6] = $2;
    }

/^r/ && /AGT/ && /cbr/ {
    data += $8;
    packets++;
    sumDelay += $2 - sendtimes[$6];
    t_end = $2;
}

/^s/ && /_0_/ && /AODV/ {
    if (rt_start == "")
        rt_start = $2;
}

/^r/ && /_0_/ && /AODV/ {
    rt_end = $2;
}

END {
    printf("Total data received\t: %d Bytes\n", data);
    printf("Total packets received\t: %d\n", packets);
    printf("CBR start\t\t: %s\n", t_start);
    printf("CBR end \t\t: %s\n", t_end);
    printf("Mean CBR packet delay\t: %f sec\n", (1.0 * sumDelay) / packets);
    printf("Routing start\t\t: %s\n", rt_start);
    printf("Routing end\t\t: %s\n", rt_end);
    printf("Routing delay\t\t: %f sec\n", rt_end - rt_start);
}

```

3.3 Εκτέλεση του προγράμματος ανάλυσης

Για την εκτέλεση προγραμμάτων awk χρησιμοποιείται ο διερμηνέας (interpreter) awk με παραμέτρους το όνομα του αρχείου που περιγράφει τις διαδικασίες της ανάλυσης και το όνομα του αρχείου που περιλαμβάνει τα δεδομένα που θα αναλυθούν. Εάν ο παραπάνω κώδικας έχει αποθηκευτεί στο αρχείο **lab8.awk** και τα δεδομένα βρίσκονται στο αρχείο **lab8.tr**, τότε εκτελούμε την εντολή:

```
awk.exe -f lab8.awk < lab8.tr
```

Αυτή η εντολή θα εκτυπώσει στην οθόνη τα ακόλουθα: το πλήθος των δεδομένων και το συνολικό αριθμό των πακέτων κίνησης CBR που ελήφθησαν, το χρόνο αποστολής του πρώτου πακέτου κίνησης CBR από τον αποστολέα, το χρόνο λήψης του τελευταίου πακέτου κίνησης CBR από τον παραλήπτη, τη μέση καθυστέρηση μετάδοσης πακέτου, το χρόνο αποστολής του πρώτου πακέτου AODV από τον αποστολέα, το χρόνο λήψης του τελευταίου πακέτου AODV από τον αποστολέα, και το χρόνο

συμπλήρωσης των πινάκων δρομολόγησης.

Για να αποθηκεύσουμε τα αποτελέσματα της ανάλυσης στο αρχείο **results.txt** εκτελούμε την εντολή

```
awk.exe -f lab8.awk < lab8.tr > results.txt
```

Προσοχή: Σε περιβάλλον MS Windows ο διερμηνέας awk χρησιμοποιεί ως σύμβολο υποδιαστολής (decimal mark), το σύμβολο που είναι καθορισμένο από τα *regional settings* της εγκατάστασης. Σε πολλές περιπτώσεις αυτό το σύμβολο είναι το κόμμα (,) και όχι η τελεία (.), με αποτέλεσμα το awk να μην μπορεί να χειριστεί σωστά πραγματικούς αριθμούς από το αρχείο εισόδου. Για να αντιμετωπιστεί το πρόβλημα αυτό, έχουμε προσθέσει τον ακόλουθο κανόνα ο οποίος αντικαθιστά όπου χρειάζεται τις τελείες με κόμματα.

```
{  
    # Apply dm fix if needed  
    if (dmfix) sub(/\./, ",", $0);  
}
```

Εναλλακτικά, θα μπορούσαμε να ορίσουμε την τελεία ως το σύμβολο για την υποδιαστολή αλλάζοντας τις ρυθμίσεις των Windows (Control Panel → Regional and Language Options).

4. Μελέτη συμπεριφοράς ασύρματου τοπικού δικτύου IEEE 802.11

4.1 Επίδραση απόστασης κόμβων

Εκτελέστε σειρά προσομοιώσεων με δύο κόμβους (θέτοντας την παράμετρο *opt(nn)* ίση με 2) μεταβάλλοντας την οριζόντια απόστασή τους (αλλάζοντας την τιμή της παραμέτρου *opt(distx)*), προσαρμόζοντας ανάλογα την παράμετρο *opt(gridx)* που καθορίζει το μήκος του πλέγματος, όπου απαιτείται. Πρέπει να πραγματοποιηθούν τουλάχιστον δέκα προσομοιώσεις.

- (α) Για αυτές τις προσομοιώσεις και με τη βοήθεια του awk script να γίνει γραφική παράσταση της μέσης τιμής της καθυστέρησης ανά πακέτο συναρτήσει της απόστασης των δύο κόμβων.
- (β) Ποια είναι η ελάχιστη ακέραια τιμή της παραμέτρου *opt(distx)* για την οποία είναι αδύνατη η μεταφορά δεδομένων μεταξύ των δύο κόμβων; Ποια είναι η εμβέλεια του δικτύου για τους δύο κόμβους;
- (γ) Τέλος, να γίνει γραφική παράσταση της τιμής της χρησιμοποίησης του καναλιού (utilisation) συναρτήσει της απόστασης των δύο κόμβων. Τι παρατηρείτε;

4.2 Επίδραση αριθμού κόμβων

Στη συνέχεια θα γίνει σειρά προσομοιώσεων με μεταβλητό πλήθος κόμβων και θα μελετηθεί η επίδραση του πλήθους των κόμβων στη χρησιμοποίηση του καναλιού και στη μέση καθυστέρηση. Για το σκοπό αυτό θα γίνει σειρά προσομοιώσεων με σταθερή απόσταση διαδοχικών κόμβων ίση με 20 m και τοποθετώντας διαδοχικά 2, 4, 8, 12, 16, 20, 26 και 32 κόμβους πάνω στον άξονα x.

- (α) Υπολογίστε με τη βοήθεια του awk script τη μέση τιμή της καθυστέρησης ανά πακέτο για κάθε μία από αυτές τις τοπολογίες, προσαρμόζοντας ανάλογα την παράμετρο *opt(gridx)* που καθορίζει το μήκος του πλέγματος, όπου απαιτείται. Πώς επηρεάζει τη μέση καθυστέρηση το πλήθος των κόμβων και γιατί; Σχεδιάστε διάγραμμα που απεικονίζει τη μέση καθυστέρηση μετάδοσης

πακέτου (σε msec) συναρτήσει του πλήθους των κόμβων της τοπολογίας.

- (β) Πόσο χρόνο διαρκεί η συμπλήρωση των πινάκων δρομολόγησης σε κάθε μία από τις παραπάνω τοπολογίες; Σχεδιάστε διάγραμμα που απεικονίζει το χρόνο συμπλήρωσης των πινάκων δρομολόγησης (σε msec) συναρτήσει του πλήθους των κόμβων της εκάστοτε τοπολογίας.
- (γ) Σχεδιάστε το πλήθος των βημάτων (hops) που απαιτούνται για τη μετάδοση των πακέτων κίνησης CBR από τον αποστολέα στον παραλήπτη συναρτήσει του πλήθους των κόμβων της εκάστοτε τοπολογίας. Πώς δικαιολογείται η συμπεριφορά αυτή του δικτύου;